
.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VC     0      DC     10
VAC1      VIN    0      DC     0      SIN(   0      1   1000 )
I1        VEN1   0      1u
QN1       VCN1   VBN1   VEN1   NPN1   1.00
QN2       VCN2   0      VEN1   NPN1   1.00
R1        VCN1   VC     52K
R2        VCN2   VC     52K
E_GAIN1   OUT    0      VCN1   VCN2   1
E_GAININ  VBN1   0      VIN    0      1m

.control
tran            25u                     5m         0    1u
plot            out
echo            "THD% versus VIN_vpk and Temp_C"
setplot         new
set             NameList =  ( minus55C plus25C plus125C )
compose         TempVals    values  (-55)  25    125
compose         VinVals     values   1m 3m 10m 30m 100m 300m
settype         voltage     VinVals
let             NoOfTemp =  length(TempVals)
let             NoOfVin =   length(VinVals)

begin
unset            interrupt
* ===============Loop_Temp============
let              j         =  1
while            (j       <=  NoOfTemp )
let              Tmp       =  TempVals[j-1]
set              temp      =  $&Tmp
set              thisName  =  $NameList[$&j]
let              $thisName =  0 * vector(NoOfVin)
echo             "$&j $&Tmp   $temp $thisName "

* ===============Loop_Vin==============
let              k             = 1
while            (k           <= NoOfVin )
let              Vin           = VinVals[k-1]
alter            e_gainin gain = $&Vin
tran             25u              5m        0    1u
linearize
set             specwindow=   "blackman"
spec            200    8k      200     v(out)
let             thdsq =mag(out[9])^2 +mag(out[14])^2  +mag(out[19])^2  +mag(out[24])^2
let             thd_percent= 100*sqrt(thdsq)/mag(out[4]) 
echo            "$&unknown.Vin       $&thd_percent"
let             unknown.{$thisName}[unknown.k-1] = thd_percent
repeat          3
destroy
end
if               ($?interrupt)
goto             bail
endif
let              k =          k + 1



endwhile
setscale         VinVals
plot             $NameList loglog title "THD_% vs Vin_pK and Temp_C"
let              j =          j + 1
endwhile
label            bail
echo            "Done."
end
.endc

*======== ====== ====== ====== ====== ====== ====== ====== ======
.model    NPN1   NPN(   BF=2100 VAF=216 )
.model    PNP1   PNP(   BF=2100  VAF=21)
.end

=====================END_OF_SPICE=======================

This simulation only works on MacSpice for now.
Data in spice apparently gets stored in vectors which
are ready to be plotted. 

This is a classical plot of the distortion versus differential
input voltage. The distortion often limits the practical magnitude 
of the input signal to less than ideal signal to noise ratio levels.
For instance the 1% distortion level is at a 18mV peak level. 
At this input voltage, the output current is at 35% of the 
maximum available output.

The distortion simulation does a transient analysis at various 
input levels and temperatures. The distortion is found by 
doing an RMS sum of the harmonics and then dividing by the 
fundamental.

Plotting distortion versus input level shows that
the distortion is constant in shape and is 
scaled to absolute temperature.


